
NCL 绘图示例（六）： 

中国区域内的站点（客观分析）图 

施宁 

（南京信息工程大学 大气科学学院） 

 

begin 

;--1, 读入 2016 年 7 月中国 839 站逐日地表气温 

  temp=asciiread("./temp-839-201607.txt",-1,"integer")  ;全部读入为 1 维数组 

  var = reshape(temp,(/839,31,13/))        ;转换为 839 站 X31 天 X13 列 

  var@_FillValue = 32766 

   

  val_mon  = dim_avg_n(var(:,:,7),1)/10. ;计算 839 站的月平均气温 

  lat_stat = var(:,0,1)/100.             ;每个站点的纬度 

  lon_stat = var(:,0,2)/100.             ;每个站点的经度   

   

;--2 进行客观分析，则将台站资料插值至等经纬度网格网格点上 

   delta = 0.5   ;网格距 

   lon_s = 70 

   lon_e = 140 

   lat_s = 15 

   lat_e = 55 

   nlat = ceil((lat_e-lat_s)/delta) + 1 ;29 

   nlon = ceil((lon_e-lon_s)/delta) + 1 ;21   

   glat = fspan(lat_s,lat_e,toint(nlat))    



   glon = fspan(lon_s,lon_e,toint(nlon))  

   ;-设定经纬度的单位  

   glon!0          = "lon" 

   glon@long_name  = "longitude" 

   glon@units      = "degree_east" 

   glat!0          = "lat" 

   glat@long_name  = "latitude" 

   glat@units      = "degree_north" 

   ;客观分析  

   rscan=(/5.,2.5,1/) ;不超过 10 个数，通常 1 至 4 个数。数值依次递减，表示每

次迭代分析的半径（单位为度）    

   val_mon := 

obj_anal_ic_Wrap(lon_stat,lat_stat,val_mon,glon,glat,rscan,False);creanm 插

值;creanm 插值 

   

;--3，图形文件设置  

  plot = new(1,"graphic") 

  plot2 = new(1,"graphic")   

   

  wks  = gsn_open_wks("eps","plot-china-station") 

  gsn_define_colormap(wks,"MPL_Spectral") 

 

;--4，首先绘制中国区域内的变量   

  res=True 

  res@gsnDraw             = False 

  res@gsnFrame            = False 

  res@gsnAddCyclic        = False 

   

  res@cnFillOn            = True      

  res@cnLineColor         = "white" 

  res@cnLineThicknessF    = 0.5 

  res@cnFillDrawOrder     = "PreDraw"  

  res@cnLineDrawOrder     = "PreDraw"  

  res@gsnSpreadColorEnd   = -6 

       

  res@cnLevelSelectionMode  = "ManualLevels" 

  res@cnMinLevelValF        = 8           

  res@cnMaxLevelValF        = 32   

  res@cnLevelSpacingF       = 2 

   

  res@mpDataBaseVersion     = "Ncarg4_1"                 

  res@mpDataSetName         = "Earth..4"   

       

  res@mpFillOn               = True   ;必须设置为 True 



  res@mpOceanFillColor       = 0                              

  res@mpInlandWaterFillColor = 0   

  res@mpLandFillColor        = 0 

  res@mpAreaMaskingOn        = True  

  res@mpMaskAreaSpecifiers   = (/"China","Taiwan","Arunachal 

Pradesh","Disputed area between India and China"/) ;这才是正确的中国陆地领土范

围 

  res@mpOutlineOn            = False ;不绘制国界线，后面将通过 shapefile

文件绘制正确的国界线 

   

  re = res   ;复制上述 res 属性，为绘制南海地区做准备 

     

  res@mpMinLatF             = 15   

  res@mpMaxLatF             = 60   

  res@mpMinLonF             = 70  

  res@mpMaxLonF             = 140      

  res@mpGridAndLimbOn       = True  

  res@mpGridLineColor       = "black" 

  res@mpGridLatSpacingF     = 5 

  res@mpGridLonSpacingF     = 5 

  res@mpGridLineDashPattern = 16 

  res@mpGridLineThicknessF  = 0.2 

  res@gsnMajorLonSpacing    = 5 

   

  ;--设置坐标标签 

  res@tmXBMode = "Explicit" 

  lon_value = ispan(70,140,10) 

  lon_lable = lon_value+"~S~o~N~E" 

  res@tmXBValues = lon_value 

  res@tmXBLabels = lon_lable 

  lat_value = ispan(20,60,10) 

  lat_lable = lat_value+"~S~o~N~N" 

  res@tmYLValues = lat_value 

  res@tmYLLabels = lat_lable 

   

  ;绘图       

  plot = gsn_csm_contour_map(wks,val_mon,res)    

   

;--6，绘制长江、黄河及国界线 

  f=addfile("./lonlat-2river-guo.nc","r") ;长江与黄河的经纬度 

  lon_cj = f->lon_cj 

  lat_cj = f->lat_cj  

  lon_hh = f->lon_hh 

  lat_hh = f->lat_hh 



  

  pres                  = True   

  pres@gsLineColor      = "black" 

  pres@gsLineThicknessF = 1.5 

  dumhh = gsn_add_polyline(wks,plot,lon_hh,lat_hh,pres) 

  dumcj = gsn_add_polyline(wks,plot,lon_cj,lat_cj,pres)   

 

  ;通过 shapefile 文件绘制正确的中国国界线  

  shp_path= "./shp/china.shp" 

  pres@gsLineColor      = "brown" 

  pres@gsLineThicknessF = 1.0 

  outline_china = gsn_add_shapefile_polylines(wks,plot,shp_path,pres) 

   

  draw(plot)   

     

;--7，绘制南海  

  ;首先在图形右下方绘制一个白色多边形 

  xlon = (/130,140,140,130,130/) 

  ylat = (/15,15,27.5,27.5,15/) 

  reg =True 

  reg@gsFillColor = "white" 

  gsn_polygon(wks,plot,xlon,ylat,reg) 

    

  ;再在该多边形内绘制南海区域 

  re@vpXF = 0.75 

  re@vpYF = 0.4 

  re@vpWidthF = 0.1 

  re@vpHeightF = 0.125 

   

  re@mpMinLatF = 0   

  re@mpMaxLatF = 25 

  re@mpMinLonF = 105   

  re@mpMaxLonF = 125   

   

  re@mpOutlineOn  = True    ;此时可设置为 True 

 

  re@tmXBLabelsOn = False 

  re@tmXTLabelsOn = False   

  re@tmYLLabelsOn = False 

  re@tmYRLabelsOn = False    

  re@tmXBOn = False 

  re@tmXTOn = False   

  re@tmYLOn = False 

  re@tmYROn = False  



  re@lbLabelBarOn = False 

   

  ;绘制南海地区 

  plot2 = gsn_csm_contour_map(wks,val_mon,re)  

  outline_scs = gsn_add_shapefile_polylines(wks,plot2,shp_path,pres) 

  draw(plot2)   

   

  frame(wks) 

end 

 

 


